<commit_before>@ECHO OFF
REM Some variables
SET "EMBEDDED_DIR=%~dp0\..\embedded"

REM Save some variables we're going to change so we can restore them later
SET "VAGRANT_GEM_HOME_SAVED=%GEM_HOME%"
SET "VAGRANT_GEM_PATH_SAVED=%GEM_PATH%"
SET "VAGRANT_GEMRC_SAVED=%GEMRC%"
SET "VAGRANT_PATH_SAVED=%PATH%"
SET "VAGRANT_RUBYOPT_SAVED=%RUBYOPT%"

REM Set environmental variables
SET "GEM_HOME=%EMBEDDED_DIR%\gems"
SET "GEM_PATH=%GEM_HOME%"
SET "GEMRC=%EMBEDDED_DIR%\etc\gemrc"

REM Export an enviromental variable to say we're in a Vagrant
REM installer created environment.
SET "VAGRANT_INSTALLER_ENV=1"
SET "VAGRANT_INSTALLER_EMBEDDED_DIR=%EMBEDDED_DIR%"
SET "VAGRANT_INSTALLER_VERSION=<%= @installer_version %>"

REM Prepend embedded bin to PATH so we prefer those binaries
SET "PATH=%EMBEDDED_DIR%\bin;%EMBEDDED_DIR%\gnuwin32\bin;%PATH%"

REM Remove RUBYOPT, which causes serious problems.
SET RUBYOPT=

REM Run Vagrant...
"%EMBEDDED_DIR%\..\embedded\bin\ruby.exe" "%EMBEDDED_DIR%/../embedded/gems/bin/%~n0" %*

REM Store the exit status so we can re-use it later
SET "VAGRANT_EXIT_STATUS=%ERRORLEVEL%"

REM Restore some environmental variables we changed
SET "GEM_HOME=%VAGRANT_GEM_HOME_SAVED%"
SET "GET_PATH=%VAGRANT_GEM_PATH_SAVED%"
SET "GEMRC=%VAGRANT_GEMRC_SAVED%"
SET "PATH=%VAGRANT_PATH_SAVED%"
SET "RUBYOPT=%VAGRANT_RUBYOPT_SAVED%"

REM Exit with the proper exit status from Vagrant
exit /b %VAGRANT_EXIT_STATUS%
<commit_msg>fixed typo in vagrant.bat to restore GEM_PATH<commit_after>@ECHO OFF
REM Some variables
SET "EMBEDDED_DIR=%~dp0\..\embedded"

REM Save some variables we're going to change so we can restore them later
SET "VAGRANT_GEM_HOME_SAVED=%GEM_HOME%"
SET "VAGRANT_GEM_PATH_SAVED=%GEM_PATH%"
SET "VAGRANT_GEMRC_SAVED=%GEMRC%"
SET "VAGRANT_PATH_SAVED=%PATH%"
SET "VAGRANT_RUBYOPT_SAVED=%RUBYOPT%"

REM Set environmental variables
SET "GEM_HOME=%EMBEDDED_DIR%\gems"
SET "GEM_PATH=%GEM_HOME%"
SET "GEMRC=%EMBEDDED_DIR%\etc\gemrc"

REM Export an enviromental variable to say we're in a Vagrant
REM installer created environment.
SET "VAGRANT_INSTALLER_ENV=1"
SET "VAGRANT_INSTALLER_EMBEDDED_DIR=%EMBEDDED_DIR%"
SET "VAGRANT_INSTALLER_VERSION=<%= @installer_version %>"

REM Prepend embedded bin to PATH so we prefer those binaries
SET "PATH=%EMBEDDED_DIR%\bin;%EMBEDDED_DIR%\gnuwin32\bin;%PATH%"

REM Remove RUBYOPT, which causes serious problems.
SET RUBYOPT=

REM Run Vagrant...
"%EMBEDDED_DIR%\..\embedded\bin\ruby.exe" "%EMBEDDED_DIR%/../embedded/gems/bin/%~n0" %*

REM Store the exit status so we can re-use it later
SET "VAGRANT_EXIT_STATUS=%ERRORLEVEL%"

REM Restore some environmental variables we changed
SET "GEM_HOME=%VAGRANT_GEM_HOME_SAVED%"
SET "GEM_PATH=%VAGRANT_GEM_PATH_SAVED%"
SET "GEMRC=%VAGRANT_GEMRC_SAVED%"
SET "PATH=%VAGRANT_PATH_SAVED%"
SET "RUBYOPT=%VAGRANT_RUBYOPT_SAVED%"

REM Exit with the proper exit status from Vagrant
exit /b %VAGRANT_EXIT_STATUS%

{'fraction_non_alphanumeric': 0.10496311200536552, 'fraction_numerical': 0.00335345405767941, 'mean_word_length': 5.748868778280543, 'pattern_counts': {'":': 0, '<': 5, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2273751', 'n_tokens_mistral': 1295, 'n_tokens_neox': 1229, 'n_words': 270}